
Bonjour nal,

Aujourd'hui je vais te parler de résolution naïve d'un problème combinatoire, en explorant un arbre. Le problème vient d'un jeu de société, et la résolution se fera en Haskell, illustrant des notions intéressantes : Anamorphisme et Deforestation_(computer_science).

Explication du jeu


Le jeu du Ricochet Robots est un jeu de société constitué d'une grille de jeu, comportant des cases, avec des murs certains côtés, et certaines cases ayant un symbole d'une certaine couleur. 


Quatre Robots (chacun d'une couleur différente) sont placés sur la grille à leur emplacement de départ.


Un objectif est tiré au hasard, il s'agit d'amener le robot désigné, sur la case désignée.


Pour amener le robot sur cette case, les joueurs peuvent utiliser tous les robots, et les déplacer sur la grille librement, en respectant « les mouvements des robots » : 



Les robots se déplacent en ligne droite jusqu'à rencontrer un obstacle (mur/autre robot)




Les joueurs tentent de trouver une manière d'amener le robot sur la case. Le premier qui en trouve une annonce le nombre de coups  nécessaires. Un chronomètre est alors déclenché et permet aux autres joueurs de disposer de 50 secondes pour améliorer le nombre de coups annoncés. 


À la fin des 50 secondes, le joueur avec le moins de déplacements remporte le point et bouge effectivement les pièces sur le plateau, et un autre objectif est tiré.

Évaluation des stratégies


[image: Un exemple de résolution]

En avant


Un être humain va plutôt se concentrer sur les moyens de bouger la pièce concernée par l'objectif, et utiliser les autres comme « cales », afin de permettre certains mouvements. Cette méthode est compliquée à expliquer et théoriser, on va donc en extraire l'essence la plus simple.


On dit qu'à partir d'une configuration donnée, on regarde les mouvements possibles de toutes les pièces, et sélectionne les branches qui semblent mener à une combinaison plausible (ce qui vient avec l'expérience), ce en partant principalement sur des branches avec la pièce concernée par l'objectif.


L'arbre des mouvements possibles est un arbre de facteur de branchement 16, puisqu'à chaque étape il est possible de faire 16 mouvements (sans compter les mouvements inutiles à cause des murs) : 4 pièces et 4 directions sont possibles.

En arrière


Une autre stratégie est de regarder d'où peut provenir une pièce qui arriverait sur la case voulue, et chercher des mouvements plausibles (en mettant des cales, etc …) pour y arriver. Cette méthode est un peu moins intuitive à coder, mais pour un être humain, il est parfois « facile » de déterminer d'où peut probablement provenir la pièce, et remonter ainsi le chemin vers une case qui est plus facilement accessible.


Néanmoins, il faut noter que le joueur humain oscille souvent entre les deux méthodes, l'une permettant d'avancer, l'autre de simplifier le problème.

Exploration naïve vers l'avant


Le choix le plus simple est celui de la résolution en avant, puisqu'une description sous forme d'arbre est directement disponible.


L'idée est donc de faire une exploration intelligente de l'arbre, de manière à atteindre le plus facilement possible une solution, et de préférence optimale. Les deux types de parcours les plus courants sont :



	Le parcours en largeur : efficace en temps 

	Le parcours en profondeur : efficace en espace 


Le parcours en largeur permet de trouver facilement une solution optimale le plus rapidement possible. Contrairement au parcours en profondeur, qui ne trouvera pas directement une solution avec le moins de coups possible.

Définition de l'arbre


On commence par définir un arbre sans information dans les nœuds, et avec étiquettes de transitions, de la manière la plus simple possible :


data Tree a = N [(a, Tree a)]

Construction d'un arbre


Si on regarde le type de construction, on se rend compte que 


N :: [(a, Tree a)] -> Tree a


Si on « généralise le type » on arrive à une fonction à deux paramètres :


data Tree' a b = N' [(a, b)]
N' :: [(a, b)] -> Tree' a b


On a la correspondance Tree a = Fix (Tree' a) pour ceux qui connaissent cet opérateur. En effet, on se rend compte que : 


Tree a = Tree' a (Tree a) = Tree' a (Tree' a (Tree a)) ...


L'idée est donc de « faire pousser un arbre » avec l'opérateur N' appliqué un certain nombre de fois. La fonction de « pousse » aura donc le type :


pousse :: (b -> [(a, b)]) -> b -> Tree a


Prenant une fonction qui va générer un nœud de type Tree' à partir d'une graine b, et d'une graine initiale.


Pour des raisons de types, on n'arrivera jamais à transformer le type Tree' en Tree avec un nombre « fini » d'étapes, donc il faut changer de stratégie, et utiliser une définition récursive pour notre fonction pousse : 


pousse branche graine = N $ map traiteSousBranche liste 
    where
        liste                            = branche graine 
        traiteSousBranche (x,sousGraine) = (x, pousse branche sousGraine)

Fonction de construction


Maintenant que l'on a un moyen de faire pousser naturellement des arbres, on peut se demander si cette fonction est assez générique pour construire le seul arbre qui nous intéresse ici : l'arbre des déplacements !


On commence par donner les deux trois choses importantes :


data Piece = Rouge | Vert | Bleu   | Gris deriving (Eq,Show,Read,Enum)
data Depl  = Haut  | Bas  | Droite | Gauche deriving (Eq,Show,Read,Enum)

data Move = Move Piece Depl deriving (Eq,Show,Read)


Ensuite, on peut se demander quelles sont les options à chaque branche :


options :: [Move]
options =  [ Move p d | p <- [Rouge .. Gris],
                        d <- [Haut  .. Bas ] ]


Maintenant une question se pose : quel est le type de notre fonction de construction ? 


On sait déjà qu'elle doit avoir une signature dans la famille b -> [(a,b)], avec b le type d'une graine qui génère une branche, et a les étiquettes qui seront vraiment dans l'arbre. 


L'arbre que l'on veut construire est du type Tree Move, donc a = Move, la question est donc : mais que « vaut » b ? En réalité, n'importe quel type convient, car à chaque branchement, on a exactement les mêmes mouvements possibles (comme expliqué au début) : aucune graine n'est nécessaire !


Le type sera donc rien : b = ().


arbreFunc :: () -> [(Move, ())]
arbreFunc _ = map (flip (,) ()) options


Remarque : l'arbre que l'on construit ici est infini parce qu'à 

chaque étape on ajoute de nouvelles branches. Si on veut un arbre 

fini, il faut que pour chaque branche, on arrive un jour à une liste vide … 

Parcours en largeur


L'idée du parcours en largeur est la suivante : à chaque étape, je parcours les nœuds d'un étage. Quand je le fais, je note les nœuds de l'étage suivant, ce qui me permet de savoir quel étage traiter à l'étape suivante. Pour cela on utilise souvent une file, en ajoutant au fur et à mesure les fils.


Ici, on va procéder différemment, on ne va pas coder un parcours en profondeur, mais simplement une fonction, qui va « suivre » certaines branches. On veut descendre d'un niveau dans l'arbre, en gardant pour chaque branche une information additionnelle : l'historique des modifications, et la disposition du plateau qui en résulte (pour ne pas la recalculer à chaque fois). Le type de donnée que l'on veut est alors tout simplement :


branchesSuivies :: [ (b, Tree a) ]


Avec b une information additionnelle (ici l'historique des mouvements pour simplifier).


Pour passer au niveau suivant, c'est alors très simple, mais il faut savoir ce que l'on veut. On veut aussi modifier l'historique, donc on veut une fonction du type :


nouvellesBranchesSuivies :: (b, Tree a) -> [ (b, Tree a) ]


Cette fonction permettant de dire comment ajouter les nouvelles branches qui sont suivies. Le type de la fonction « suivre » est alors tout simple 


suivre :: ((b,Tree a) -> [(b,Tree a)]) -> [(b, Tree a)] -> [(b, Tree a)]


Cette signature un peu compliquée peut se généraliser facilement pour mieux en comprendre l'essence :


suivre :: (c -> [c]) -> [c] -> [c]


On dispose alors de fonctions très génériques qui vont faire le travail à notre place :


map :: (a -> b) -> [a] -> [b]
concat :: [[a]] -> [a]


Une simple analyse nous montre alors que :


l :: [c]
f :: (c -> [c])

map f l :: [[c]]

concat (map f l) :: [c]


Par conséquent un candidat évident (et qui est la bonne manière de faire en plus) est  


suivre f = concat . map f

Notre fonction de suivi


Reste maintenant à écrire notre fonction de suivi de branches, c'est relativement simple, puisque l'on sait ce que l'on veut :



	On veut un historique [Move]


	L'arbre est de type Tree Move



On code donc assez naturellement la fonction 


suivreBranches :: [Move] -> Tree Move -> [([Move], Tree Move)]
suivreBranches historique (N l) = map (\(x,y) -> (x : historique, y)) l

Mettre de l'ordre dans tout cela


Où en est-on de la résolution du problème ? Et bien en réalité on a presque terminé … Enfin, il reste seulement une étape (pénible et très peu théorique) : à chaque fois que l'on descend d'un niveau, il faut vérifier si on a bien une solution gagnante.


Pour cela il faut :



	Avoir une grille (la construire)

	Savoir déplacer les robots dessus (ie : calculer les nouvelles positions)

	Avoir un objectif

	Faire que tout marche bien … 


Ce n'est pas dur, simplement long.

Mise en perspective


On peut remarquer qu'à partir du moment où on sait déjà bouger des robots sur la grille, on sait si des mouvements sont impossibles (ie : ne bougent en réalité pas le pion), ce qui permet de tuer un certain nombre de branches directement. Les robots ne pouvant se stopper que contre un obstacle, il y a très souvent une des directions qui n'est pas possible, ce qui fait tomber de « branching-factor » à 12 (ce qui est non négligeable !).


Ainsi, plutôt que de construire un arbre Tree Move on peut construire un arbre de type Tree ([Move], Placement) qui permet à notre fonction qui fait pousser l'arbre d'avoir accès au placement, et donc de tuer directement les branches inutiles (et permet d'avoir une fonction d'aplatissement encore plus simple !).


De même, on peut remarquer que construire l'arbre est inutile, et qu'on peut le faire pousser en même temps que l'on regarde si on trouve des solutions ! À ce moment là, on comprend pourquoi le type list est aussi dit « monade non-déterministe » : 


options :: [Move]

cheminsPossibles :: ([Move], Placement) -> [([Move], Placement)]

avancer :: [([Move], Placement)] -> [([Move], Placement)]
avancer = concat . map cheminsPossibles 

trouveSolution :: [([Move], Placement)] -> Maybe ([Move], Placement)

resolution l = case trouveSolution l of 
                    Nothing -> resolution (avancer l)
                    Just s  -> s


Ce qui est un code plus concis où l'arbre intermédiaire a été retiré. On parle ici de « déforestation », c'est à dire suppression des structures intermédiaires de calcul dans un programme.

